Q:

Integrate full derivative

I want mathematica to do  
Integrate[D[f[x],x], {x,0,t}]

and achieve the answer f[t]-f[0]. But Mathematica doesn't intergrate it.
Any ideas?

A:

I'd use this nice method by Jens how to simplify symbolic integration  (notice, need the SetAttributes for version 10)
Clear[f, ff];
f /: Integrate[f[x_], x_] := ff[x];
SetAttributes[ff, {NumericFunction}]
Integrate[f[x], {x, 0, t}] /. ff -> f

A:

Mathematica solves the equivalent differential equation:
y[t] /. First@DSolve[{y'[t] == f'[t], y[0] == 0}, y, t]
(*
  -f[0] + f[t]
*)

A:

Another approach is to add an assumption, in the form of an upvalue for the limit of f, that makes f continuous:
Block[{f},
 f /: Limit[f[x_], x_ -> c_, ___] := f[c];
 Integrate[D[f[x], x], {x, 0, t}]
 ]
(*  -f[0] + f[t]  *)

It works because Integrate evaluates the antiderivative by means of directional limits at the end points.  This observation, perhaps, indicates why Mathematica does not automatically apply the Fundamental Theorem of Calculus to the integral: It is not valid in all cases.  In fact, the result above is not valid in all cases, since it yield zero for the following positive (and divergent!) integrand:
% /. f -> (1/(2 # - t)^2 &)
(*  0  *)

